

Supersedes:

5345A-19

HP MODEL 5345A ELECTRONIC COUNTER**All Serial Prefixes****HP-IB VERIFICATION PROGRAM FOR 5345A OPTION 011
USING THE HP 85A CONTROLLER**

The HP 85A program listed in the table exercises the 5345A through its various operating modes via the HP-IB Interface (Option 011). If the 5345A successfully completes all phases of the verification program, then there is a very high probability that the Option 011 Interface is working properly.

To perform the verification, set up the 5345A as shown.




Figure 1. 5345A HP-IB Verification Set-Up

E/OF/WN

1/82-02—VGM



For more information contact a local Hewlett-Packard Office. (Hewlett-Packard has 200 Sales and Service Offices in 75 countries).
Or write: Hewlett-Packard, 1501 Page Mill Road, Palo Alto, California 94304, U.S.A.

Printed in U.S.A.

Perform the following steps:

1. Key-in the program listed in *Table 1* or insert cassette, HP Part Number 59300-10002 into the HP 85A.
2. Load and run file "Autost" (type: 'CHAIN "Autost"'); press 'END LINE'.
3. Type in model number of the instrument to be tested (e.g., "5345A1"), then press 'END LINE'.
4. Set up counter as described in the display.
5. Set the ADDRESS SWITCHES to any value between 00 and 30 inclusive, except the controller address (decimal 21).
6. Always press 'CONT' to advance program.

At the end of the HP-IB Verification Program, the controller will ask if you want to repeat one of the test. Answer "1" for YES or "0" for NO, then press 'END LINE'. If YES, the controller then asks which test is to be repeated. Enter the appropriate number for the test needed (as indicated on the controller display), then press 'END LINE'. The selected test will then be repeated. At the end of the particular test, the question of whether or not to repeat a test is asked again. If you want the same test, enter "1" and press 'END LINE'; then enter the test number.

Use the following procedure if it is desired to test a specific check-point with the HP-IB Verification Program.

1. Load the tape in the normal manner and proceed until the controller displays the set up information.
2. Type: "CONT 3990"; then press 'END LINE'. The program will advance to the end and ask if any of the tests need to be repeated. Then use the method described above.

Table 1. 5345A HP-IB Verification Program Listing

```

10 ! 5345A HP-IB VERIFICATION PROGRAM ***** "5345A(011) *****
30 ENABLE KBD 1
50 DIM F$(50),B$(32),C$(32),D$(29),G$(90),H$(90),O$(32)
70 F$="          CHECK POINT #"
90 B$="-----"
110 C$="Press 'CONT' to perform test."
130 O$="Press 'CONT' when done."
150 G$="If 5345A display not correct      HP-85 displays an ERROR message."
170 D$="Verify 5345A display is:"
190 L=0 @ CRT IS 1 @ PRINTER IS 2
210 CLEAR @ DISP @ DISP @ DISP "5345A HP-IB Verification Test" @ DISP
230 DISP "***** '5345A1' *****" @ DISP @ DISP
250 DISP "After entering a 3 digit select" @ DISP
270 DISP "code (7XX), press 'END LINE'." @ DISP @ INPUT S
290 IF NOT (S=721) THEN 330 ELSE DISP "ERROR: Calculator Address !!"
310 WAIT 1000
330 IF NOT (S=721) THEN 370 ELSE DISP "calculator address=computer dump"
350 GOSUB 4690
370 IF NOT (S>730) THEN 410 ELSE DISP "out of address range->high"
390 GOSUB 4690
410 IF NOT (S<700) THEN 450 ELSE DISP "out of address range->low"
430 GOSUB 4690
450 IF NOT (S MOD 2=1) THEN 490 ELSE DISP "odd address->computer dump"
470 GOSUB 4690
490 H$="***** 1-GATE TIME TEST *****" @ CLEAR @ DISP F$;1 @ DISP H$ @ DISP
510 DISP "Each GATE TIME code is          automatically sent to 5345A."
530 DISP "Check Mode set. 5345A output is verified."
550 GOSUB 4550 @ GOSUB 4570
570 OUTPUT S ;"I2E8?I1" @ A=9 @ GOSUB 4250 @ DISP "1 SEC GATE" @ GOSUB 4650
590 OUTPUT S ;"G?I1" @ A=8 @ GOSUB 4250 @ DISP "100 MSEC GATE" @ GOSUB 4670
610 OUTPUT S ;"G>I1" @ A=7 @ GOSUB 4250 @ DISP "10 MSEC GATE" @ GOSUB 4670
630 OUTPUT S ;"G=I1" @ A=6 @ GOSUB 4250 @ DISP "1 MSEC GATE" @ GOSUB 4670
650 OUTPUT S ;"G<I1" @ A=5 @ GOSUB 4250 @ DISP "100 MICROSEC GATE"
670 GOSUB 4670
690 OUTPUT S ;"G;I1" @ A=4 @ GOSUB 4250 @ DISP "10 MICROSEC GATE"
710 GOSUB 4670
730 OUTPUT S ;"G:I1" @ A=3 @ GOSUB 4250 @ DISP "1 MICROSEC GATE"
750 GOSUB 4670
770 OUTPUT S ;"G9I1" @ A=2 @ GOSUB 4250 @ DISP "100 NSEC GATE" @ GOSUB 4670
790 OUTPUT S ;"G5I1" @ A=1 @ GOSUB 4250 @ BEEP @ DISP "MIN GATE"
810 DISP @ DISP O$ @ PAUSE
830 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
850 H$="***** 2-FUNCTION TEST *****"
870 CLEAR @ DISP F$;2 @ DISP H$ @ DISP @ DISP
890 DISP "5345A should match calculator"
910 DISP "readout otherwise an ERROR      message occurs."
930 DISP B$ @ DISP @ DISP C$ @ PAUSE @ GOSUB 4570
950 OUTPUT S ;"I2E8?G?F2I1" @ DISP "Plug-in" @ DISP
970 DISP "Verify 5345A display is:"," 00000000000 "
990 DISP @ DISP O$ @ PAUSE
1010 OUTPUT S ;"F0I1" @ A=100000000 @ GOSUB 4430
1030 GOSUB 4570 @ DISP @ DISP "Frequency" @ DISP
1050 DISP D$,"100.00000 MHz " @ DISP @ DISP O$ @ PAUSE
1070 OUTPUT S ;"F1I1" @ A=.00000001 @ GOSUB 4430 @ GOSUB 4570 @ DISP "Period"
1090 DISP @ DISP D$,"10.000000 nSEC "
1110 DISP @ DISP O$ @ PAUSE

```

```

1130 OUTPUT S ;"F5I1" @ A=1 @ GOSUB 4570 @ GOSUB 4430 @ DISP
1150 DISP "Ratio" @ DISP
1170 DISP D$, "1.0000000 " @ DISP @ DISP O$ @ PAUSE
1190 OUTPUT S ;"F3I1" @ GOSUB 4570 @ DISP "Time Interval" @ DISP
1210 DISP D$, "10.000000 nSEC" @ DISP @ DISP O$ @ PAUSE
1230 OUTPUT S ;"F4I1"
1250 GOSUB 4570 @ DISP "Start" @ DISP @ DISP D$, "Accumulating " @ DISP
1270 DISP C$[1,13]; "to stop." @ PAUSE
1290 OUTPUT S ;"F6" @ ENTER S ; A$
1310 SEND 7 ; UNL TALK 7 LISTEN 13 MLA @ BEEP
1330 GOSUB 4570 @ DISP "Stop" @ DISP
1350 DISP "Verify that 5345A has stopped accumulating. 5345A display"
1370 DISP "should be "; A$
1390 DISP @ DISP C$[1,13]; " for next "; C$[25] @ PAUSE
1410 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
1430 H$="** 3-FREQ MULT SUFFIX TEST **" @ CLEAR @ DISP F$; 3 @ DISP H$ @ DISP
1450 DISP "Each MULT SUFFIX is automatically sent to 5345A."
1470 DISP "Check mode set. Output verified"
1490 DISP "and displayed on the HP-85."
1510 GOSUB 4550 @ GOSUB 4570 @ DISP
1530 OUTPUT S ;"I2G?D;C3E?E8I1" @ A=11 @ GOSUB 4270
1550 DISP "00000000000.MHZ*" @ GOSUB 4650
1570 OUTPUT S ;"C4I1" @ A=9 @ GOSUB 4270 @ DISP "100000000.HZ*" @ GOSUB 4650
1590 OUTPUT S ;"C5" @ A=6 @ GOSUB 4270 @ DISP "100000.KHZ*" @ GOSUB 4650
1610 OUTPUT S ;"C6" @ A=3 @ GOSUB 4270 @ DISP "100.MHZ*" @ GOSUB 4650
1630 OUTPUT S ;"G5D0C7" @ A=0 @ GOSUB 4270 @ DISP ".1 GHZ"
1650 DISP @ DISP O$ @ BEEP @ PAUSE
1670 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
1690 H$="** 4-DECIMAL POINT TEST ** (FREQ MODE)"
1710 CLEAR @ DISP F$; 4 @ DISP H$ @ DISP
1730 DISP "Each DISPLAY POSITION code is automatically sent to 5345A."
1750 DISP "Check Mode set. HP-85 shows position of the dec point with"
1770 DISP "digit 10 on left of 5345A display and digit 0 on the right."
1790 DISP B$ @ DISP @ DISP C$[1,13] @ PAUSE @ GOSUB 4530 @ PAUSE @ GOSUB 4570
1810 CLEAR @ DISP H$[1,28] @ DISP
1830 OUTPUT S ;"I2G5E8?C7D1" @ A=10 @ GOSUB 4590
1850 OUTPUT S ;"D2" @ A=9 @ GOSUB 4590
1870 OUTPUT S ;"D3" @ A=8 @ GOSUB 4590
1890 OUTPUT S ;"D<" @ A=7 @ GOSUB 4590
1910 OUTPUT S ;"D=" @ A=6 @ GOSUB 4590
1930 OUTPUT S ;"D>" @ A=5 @ GOSUB 4590
1950 OUTPUT S ;"D?" @ A=4 @ GOSUB 4590
1970 OUTPUT S ;"D8" @ A=3 @ GOSUB 4590
1990 OUTPUT S ;"D9" @ A=2 @ GOSUB 4590
2010 OUTPUT S ;"D:" @ A=1 @ GOSUB 4590
2030 OUTPUT S ;"D;" @ A=0 @ GOSUB 4590 @ DISP O$ @ PAUSE
2050 OUTPUT S ;"D0"
2070 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
2090 H$="** 5-PERIOD MULT SUFFIX TEST **" @ CLEAR @ DISP F$; 5 @ DISP H$
2110 DISP @ DISP "Function=Period. Each MULT SUFFIX is automatically"
2130 DISP "sent to 5345A. Check mode set. Output verified"
2150 DISP "and displayed on the HP-85."
2170 DISP B$ @ DISP @ DISP C$[1,13] @ PAUSE @ GOSUB 4530 @ PAUSE @ GOSUB 4570
2190 OUTPUT S ;"I2F13?D1C3E8?I1" @ A=10 @ GOSUB 4250
2210 DISP ".0000000000KSEC*" @ GOSUB 4630
2230 OUTPUT S ;"C4I1" @ A=7 @ GOSUB 4290 @ DISP ".0000000100SEC*"
2250 GOSUB 4630

```

```
2270 OUTPUT S ;"C5I1" @ A=4 @ GOSUB 4290 @ DISP ".0000100000MSEC*"
2290 GOSUB 4630
2310 OUTPUT S ;"C&I1" @ A=1 @ GOSUB 4290 @ DISP ".010000000 MICROSEC"
2330 GOSUB 4630 @ OUTPUT S ;"C7D3I1" @ A=2 @ GOSUB 4270
2350 DISP "10.000000 NSEC" @ DISP @ DISP O$ @ BEEP @ PAUSE
2370 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
2390 H$="**** 6-INT/EXT GATE TEST ****" @ CLEAR @ DISP F$;6 @ DISP H$
2410 DISP @ DISP "Operation of Internal/External Gate selection checked."
2430 DISP B$ @ DISP @ DISP C$
2450 OUTPUT S ;"I2E8?G?E;I1" @ A=0 @ PAUSE @ GOSUB 4570
2470 DISP "External Gate" @ DISP
2490 DISP D$,"000000000000 " @ DISP @ DISP O$ @ PAUSE
2510 OUTPUT S ;"E3" @ A=100000000 @ GOSUB 4430 @ GOSUB 4570
2530 DISP "Internal Gate" @ DISP
2550 DISP D$,"100.00000 MHz Gating" @ DISP @ DISP O$ @ PAUSE
2570 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
2590 H$="*** 7-SAMPLE RATE TEST ***" @ CLEAR @ DISP F$;7
2610 DISP H$ @ DISP @ DISP C$ @ PAUSE @ GOSUB 4570
2630 OUTPUT S ;"I2G<E<?8"
2650 DISP "Sample rate control bypassed." @ DISP
2670 DISP D$,"Blank","and MHz & GATE annunciators are on."
2690 DISP @ DISP O$ @ PAUSE
2710 OUTPUT S ;"E4" @ GOSUB 4570
2730 DISP "Max Sample Rate" @ DISP
2750 DISP D$,"100.00 MHz Gating " @ DISP @ DISP O$ @ PAUSE
2770 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
2790 H$="**** 8-HOLD TEST ****" @ CLEAR @ DISP F$;8 @ DISP H$ @ DISP
2810 DISP "HOLD sent to 5345A. Sends three Sample Trigger Commands. Each"
2830 DISP "time the 5345a receives one of"
2850 DISP "the three Sample Trigger "
2870 DISP "Commands, the GATE annunciator"
2890 DISP "light flashes on."
2910 DISP "Sample count is displayed on HP-85."
2930 DISP B$ @ DISP @ DISP C$ @ PAUSE @ GOSUB 4570
2950 OUTPUT S ;"I2E8?G?E9I1"
2970 DISP D$,"000000000000 Hz No GATE" @ DISP @ DISP O$ @ PAUSE
2990 GOSUB 4570 @ WAIT 1000 @ OUTPUT S ;"J1"
3010 BEEP @ DISP "SAMPLE 1" @ DISP @ WAIT 2000
3030 OUTPUT S ;"J1" @ BEEP
3050 DISP "SAMPLE 2" @ DISP @ WAIT 2000
3070 OUTPUT S ;"J1" @ BEEP
3090 DISP "SAMPLE 3" @ DISP @ WAIT 2000
3110 OUTPUT S ;"E1"
3130 BEEP @ DISP "NOT HOLD--";D$,"gating" @ DISP @ DISP O$ @ PAUSE
3150 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
3170 DISP
3190 H$="***** 9-ACCUMULATE TEST *****"
3210 CLEAR @ DISP F$;9 @ DISP H$ @ DISP
3230 DISP "Checks operation of ACCUMULATE A+B and ACCUMULATE A-3 modes."
3250 DISP B$ @ DISP @ DISP C$ @ PAUSE @ GOSUB 4570
3270 OUTPUT S ;"I2F4E8?l=G?I1"
3290 DISP "A+B code sent" @ DISP
3310 DISP D$,"Accumulating" @ DISP @ DISP C$[1,13];"to stop." @ PAUSE
3330 OUTPUT S ;"F6E75" @ A=0 @ GOSUB 4430 @ GOSUB 4570
3350 DISP "A-B code sent" @ DISP
3370 DISP D$,"000000000000" @ DISP @ DISP O$ @ PAUSE
3390 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
```

```
3410 H$="***** 10-OUTPUT TEST *****" @ CLEAR @ DISP F$;10 @ DISP H$ @ DISP
3430 DISP "Checks operation of 'WAIT until addressed' and 'ONLY IF"
3450 DISP "addressd' output modes."
3470 DISP B$ @ DISP @ DISP C$ @ PAUSE @ GOSUB 4570
3490 OUTPUT S ; "I2F4E?8:15G?I1"
3510 DISP "Wait until addressed" @ DISP
3530 DISP D$, "Blank & GATE" @ DISP @ DISP O$ @ DISP @ PAUSE
3550 OUTPUT S ; "E2I1"
3570 DISP "Only if addressed" @ DISP
3590 DISP D$, "accumulating" @ DISP @ DISP C$[1,13]; "to stop."
3610 PAUSE @ GOSUB 4570
3630 DISP " **Takes 1 reading.Stops. When      CONT is pressed, takes"
3650 DISP "10 more readings and displays      them on HP-85.' Stops and"
3670 DISP "displays the final reading."
3690 DISP B$ @ DISP @ DISP G$ @ DISP
3710 OUTPUT S ; "E8F6"
3730 ENTER S ; A$
3750 SEND 7 ; UNL TALK 7 LISTEN 13 MLA
3770 DISP D$,A$ @ PAUSE @ GOSUB 4570
3790 OUTPUT S ; "I2E8?F4I1" @ X=0
3810 WAIT 1000
3830 OUTPUT S ; "F6"
3850 ENTER S ; A$
3870 SEND 7 ; UNL TALK 7 LISTEN 13 MLA
3890 DISP A$ @ BEEP @ WAIT 1000 @ X=X+1
3910 IF X>9 THEN GOTO 3950
3930 OUTPUT S ; "F4" @ GOTO 3810
3950 GOSUB 4570 @ DISP D$,A$ @ DISP @ DISP O$ @ PAUSE @ DISP @ DISP
3970 IF L=1 THEN GOTO 3990 ELSE GOSUB 4510
3990 CLEAR @ DISP @ DISP @ DISP
4010 DISP "If you need to repeat a Check", "Point, enter 1; if not, enter 0,"
4030 DISP "then press 'END LINE'." @ DISP @ INPUT L
4050 IF L<>1 AND L<>0 THEN 3990
4070 IF L=0 THEN 4190
4090 CLEAR @ DISP @ DISP @ DISP
4110 DISP "After selecting the Check Point number (1 thru 10), then,"
4130 DISP "press 'END LINE'." @ DISP @ INPUT N
4150 IF N>10 OR N<1 THEN 4110
4170 ON N GOTO 490,850,1430,1690,2090,2390,2590,2790,3190,3410
4190 CLEAR @ GOSUB 4510 @ DISP @ DISP @ DISP @ DISP @ DISP B$ @ DISP
4210 DISP "END OF 5345A (011) ELECTRONIC      COUNTER HB-IB Verification test."
4230 DISP @ DISP B$ @ WAIT 1000 @ LOCAL S @ REWIND @ END
4250 WAIT 50 @ R=69 @ C=-2 @ GOSUB 4310 @ OUTPUT S ; "I1" @ RETURN
4270 WAIT 50 @ R=46 @ C=-1 @ GOSUB 4310 @ OUTPUT S ; "I1" @ RETURN
4290 WAIT 50 @ R=49 @ C=-2 @ GOSUB 4310 @ OUTPUT S ; "I1" @ RETURN
4310 ENTER S USING "#,B" ; B
4330 IF B=R THEN GOTO 4390
4350 C=C+1
4370 GOTO 4310
4390 IF A<>C THEN DISP "ERROR C=",C;"      ";A @ DISP
4410 RETURN
4430 WAIT 50 @ ENTER S ; C @ ENTER S ; C
4450 IF A<>C THEN DISP "ERROR" @ DISP
4470 OUTPUT S ; "I1"
4490 RETURN
4510 BEEP @ WAIT 250 @ BEEP @ RETURN
4530 CLEAR @ DISP H$ @ DISP @ DISP G$ @ DISP B$ @ DISP @ DISP C$ @ RETURN
```

```
4550 DISP B$ @ DISP @ DISP G$ @ DISP @ DISP C$ @ PAUSE @ RETURN
4570 CLEAR @ DISP H$ @ DISP @ RETURN
4590 GOSUB 4250 @ BEEP @ DISP "5345A digit # ";A @ WAIT 1000 @ RETURN
4610 BEEP @ WAIT 1000 @ RETURN
4630 BEEP @ WAIT 2000 @ RETURN
4650 BEEP @ WAIT 3000 @ RETURN
4670 BEEP @ WAIT 3500 @ RETURN
4690 WAIT 2000 @ GOTO 250 @ RETURN
```

